Circuit activ de ordin I – integrator

Scopul lucrarii

In lucrarea de fata ne propunem analiza unui integrator cu AO cu condensator in bucla de reactie.



Fig .1
Integrator



Fig.2
Circuitul practic

Obs: -> In Fig1. se prezinta schema unui integrator cu AO, ideal.
->In Fig2. se prezinta schema realizata practic, numit in continuare circuit.

Caracterizarea circuitului

```
> restart:with(Syrup):
> libname:="c:\\maple/SCSlib",libname:
```

Circuit real cu rezistenta paralel

Descrierea circuitului

```
> circuit:=
"Integrator cu AO
Vin 1 0 Vin
R 1 2 R
C 2 3 C
R1 2 3
E 3 0 0 2 A
.end":
```

Calculul tensiunilor nodale si a curentilor prin laturi

```
> syrup(circuit,ac,'curenti','tensiuni'):
Syrup/parsedeck:   Analyzing SPICE deck "Integrator cu AO" (ignoring this
line)
> tensiuni;
```

$$\{v_1 = Vin,\ v_3 = -\frac{A\,R1\,Vin}{A\,R + R1 + R + A\,s\,C\,R\,R1 + s\,C\,R\,R1},\ v_2 = \frac{R1\,Vin}{A\,R + R1 + R + A\,s\,C\,R\,R1 + s\,C\,R\,R1}\}$$

```
> curenti:
```

Calcularea functiei de transfer

Exista mai multe metode de calcul a functiei de transfer:

<u>Metoda 1</u>: folosind formula amplificarii

Stim ca functia de transfer a unui asemenea circuit este descrisa de formula:

$$H(s) = -\frac{Z_2(s)}{Z_1(s)},\ Z_2(s) = \frac{R_1 * \frac{1}{sC}}{R_1 + \frac{1}{sC}},\ Z_1(s) = R$$

Astfel functia de transfer devine:

$$H(s) = -\frac{\frac{R_1 * \frac{1}{s*C}}{R_1 + \frac{1}{s*C}}}{R} = -\frac{R * \frac{1}{R_1 * C}}{s + \frac{1}{R_1 * C}}$$

Notand: a= $\frac{1}{R*C}$ si a1= $\frac{1}{R_1*C}$ va rezulta ca :

$$H(s) = -\frac{a}{s + a_1}$$

<u>Metoda 2:</u> folosind modelul nulor-norator

Cu ajutorul modelului nulor-norator al AO:



Scriind ecuatiile Kirchoff 1 si 2 pe circuit :

$$\begin{cases} V_{in} = Z_1 I \\ V_{out} = Z_2 I \end{cases} \qquad \begin{cases} V_{in} = RI \\ V_{out} = (R1 \| \frac{1}{sC}) I \end{cases}$$

Obtinem aceeasi functie de transfer obtinuta cu metoda 1, Z1 si Z2 avand aceeasi forma.

<u>Metoda 3:</u> calcul simbolic

Folosind Maple si pachetul Syrup se obtine:

```
> Hcircuit:=eval(v[3]/v[1],tensiuni);
```

$$Hcircuit := -\frac{A\,R1}{A\,R + R1 + R + A\,s\,C\,R\,R1 + s\,C\,R\,R1}$$

Calcularea functiei de transfer pentru AO ideal

```
> Hcircuitideal:=limit(Hcircuit,A=infinity);
```

$$Hcircuitideal := -\frac{R1}{R\,(1 + s\,C\,R1)}$$

Descrierea folosind ecuatiile de stare a circuitului

1.Ecuatii de stare

```
> syrup(circuit,tran,'curenti','tensiuni');
Syrup/parsedeck:   Analyzing SPICE deck "Integrator cu AO" (ignoring this
line)
```

$$\{\frac{\partial}{\partial t} v_C(t) = -\frac{R\,v_C(t) + A\,R\,v_C(t) + R1\,v_C(t) - R1\,Vin - R1\,A\,Vin}{C\,R\,R1\,(A+1)}, v_C(0) = 0\}, \{v_C(t)\}$$

2. Ecuatii de iesire

```
> tensiuni;
```

$$\{v_1 = Vin, v_2 = \frac{v_C(t)}{A+1}, v_3 = -\frac{A\,v_C(t)}{A+1}\}$$

```
> curenti:
```

Integrator

Descrierea circuitului

```
> integrator:=
"Integrator cu AO
Vin 1 0 Vin
R 1 2 R
C 2 3 C
E 3 0 0 2 A
.end":
```

Calculul tensiunilor nodale si a curentilor prin laturi

```
> syrup(integrator,ac,'curenti','tensiuni'):
Syrup/parsedeck:   Analyzing SPICE deck "Integrator cu AO" (ignoring this
line)
> tensiuni;
```

$$\{v_{Out} = -\frac{A\,Vin}{1 + s\,C\,R + A\,s\,C\,R}, v_{Ind} = \frac{Vin}{1 + s\,C\,R + A\,s\,C\,R}, v_{In} = Vin\}$$

```
> curenti:
```

Calcularea functiei de transfer

Exista mai multe metode de calcul a functiei de transfer

Metoda 1 folosind formula amplificarii

Stim ca functia de transfer a unui asemenea circuit este descrisa de formula:

$$H(s) = -\frac{Z_2(s)}{Z_1(s)},\ Z_2(s) = \frac{1}{s*C},\ Z_1(s) = G$$

Astfel functia de transfer devine:

$$H(s) = -\frac{1}{s*R*C}$$

Metoda 2: folosind modelul nulor-norator

Cu ajutorul modelului nulor-norator al AO:

Scriind ecuatiile Kirchoff 1 si 2 pe circuit :

$$\begin{cases} V_{in} = Z_1 I \\ V_{out} = Z_2 I \end{cases} \quad \begin{cases} V_{in} = RI \\ V_{out} = \frac{1}{sC} I \end{cases}$$

Obtinem aceeasi functie de transfer obtinuta cu metoda 1, Z1 si Z2 avand aceeasi forma.

Metoda 3: calcul simbolic

Folosind Maple si pachetul Syrup se obtine:

```
> H:=eval(v[3]/v[1],tensiuni);
```

$$H := -\frac{A}{1 + A\,s\,C\,R + s\,C\,R}$$

Calcularea functiei de transfer pentru AO ideal

```
> Hideal:=limit(H,A=infinity);
```

$$Hideal := -\frac{1}{s\,C\,R}$$

Descrierea folosind ecuatiile de stare a circuitului

1.Ecuatii de stare

```
> syrup(integrator,tran,'curenti','tensiuni');
Syrup/parsedeck:   Analyzing SPICE deck "Integrator cu AO" (ignoring this
line)
```

$$\{v_C(0) = 0, \frac{\partial}{\partial t} v_C(t) = \frac{-v_C(t) + Vin + A\,Vin}{R\,C\,(A+1)}\}, \{v_C(t)\}$$

2. Ecuatii de iesire

```
> tensiuni;
```

$$\{v_1 = Vin, v_2 = \frac{v_C(t)}{A+1}, v_3 = -\frac{A\,v_C(t)}{A+1}\}$$

```
> curenti:
```

Pentru circuitul real daca consideram o gama de frecvente pentru care rezistenta R1 >> Zc(?):

```
> limit(Hcircuitideal,R1=infinity);
```

$$-\frac{1}{s\,C\,R}$$

Obs: calculind functia de transfer pentru circuitul real in gama de frecventa pentru care rezistenta R1 se poate neglija s-a obtinut aceeasi relatie.

Functia de transfer H(s)

S-a calculat in sectiunea anterioara functia de transfer pentru circuitul real (cu rezistenta R1) si pentru integrator:

$$\text{integrator: } H(s)= -\frac{\alpha}{s} \text{ cu } \alpha=\frac{1}{R\,C}$$

$$\text{circuit: } H(s)= -\frac{\alpha}{s+\alpha 1} \text{ cu } \alpha=\frac{1}{R\,C},\ \alpha 1=\frac{1}{R1\,C}$$

```
> Hideal;Hcircuitideal;
```

$$-\frac{1}{s\,C\,R},\quad -\frac{R1}{R\,(1+s\,C\,R1)}$$

Polii functiei de transfer:

```
> RootOf(denom(Hideal)=0,s);RootOf(denom(Hcircuitideal)=0,s);
```

$$0,\quad -\frac{1}{C\,R1}$$

Evaluare numerica:

```
> H:=eval(Hideal,[ C=22*1E-9, R=10^3]);Hc:=eval(Hcircuitideal,[
C=22*1E-9, R=10^3, R1=10^5]);PZ[numeric](Hc,s);
```

$$H := -45454.54545\,\frac{1}{s},\quad Hc := -100\,\frac{1}{1+.002200000\,s},\quad [p1\quad -454.5]$$

Polii functiei de transfer:

```
> Bode[castig](H);Bode[faza](H); Bode[castig](Hc);Bode[faza](Hc);
```









Interpretarea functiei de transfer:

```
> plot3d(abs(eval(H,s=sigma+I*omega)),sigma=-1000..1000,omega=-
10000..10000,axes=normal,title="Reprezentarea in spatiu a
modulului f.d.t.");
plot3d(abs(eval(Hc,s=sigma+I*omega)),sigma=-1000..1000,omega=-
```

```
10000..10000,axes=normal,title="Reprezentarea in spatiu a modulului f.d.t.");
```





Functia de transfer in regim permanent H(j?)

```
>Hoideal:=subs(s=I*omega,Hideal);Hocircuitideal:=subs(s=I*omega,Hcircuitideal);
```

$$Hoideal := \frac{I}{\omega\, C\, R},\quad Hocircuitideal := -\frac{R1}{R\,(1 + I\,\omega\, C\, R1)}$$

```
> #assume(R,positive):assume(R1,positive):assume(C,positive):
>abs_H0:=abs(limit(Hocircuitideal,omega=0));arg_H0:=argument(limit(Hocircuitideal,omega=0));
abs_Hinf:=abs(limit(Hocircuitideal,omega=infinity));arg_Hinf:=argument(limit(Hocircuitideal,omega=infinity));
abs_Halpha:=abs(eval(Hocircuitideal,omega=1/(R1*C)));arg_Halpha:=argument(eval(Hocircuitideal,omega=1/(R1*C)));alpha=eval(1/(R1*C),[R=10^3, C=22*1E-9, R1=10^5]);
```

$$abs\_H0 := \left|\frac{R1}{R}\right|,\quad arg\_H0 := \mathrm{argument}\left(-\frac{R1}{R}\right),\quad abs\_Hinf := 0,\quad arg\_Hinf := 0$$

$$abs\_Halpha := \frac{1}{2}\sqrt{2}\left|\frac{R1}{R}\right|,\quad arg\_Halpha := \frac{3}{4}\pi,\quad \alpha = 454.5454545$$

```
> plot( eval( [ abs(Hoideal), abs(Hocircuitideal) ] , [R=10^3, C=22*1E-9, R1=10^5]), omega=-5000..5000,axes=normal,title="Reprezentarea modulului f.d.t. pentru regim permanent", view=[DEFAULT, -10..200]);
plot( eval( [ argument(Hoideal), argument(Hocircuitideal) ] , [R=10^3, C=22*1E-9, R1=10^5]), omega=-5000..5000,axes=normal,title="Reprezentarea modulului f.d.t. pentru regim permanent");
```





Raspuns in regim permanent

```
> restart:
> libname:="c:\\maple/SCSlib",libname:
> F:=table([dir=FOURIER,inv=inttrans[invfourier]]):
```

Determinarea raspunsului de regim permanent sinusoidal al circuitului cu AO

Pentru circuite liniare lucrând în regim permanent sinusoidal functia de transfer are semnificatia de ***amplificare generalizata.*** Un semnal $e(t)=A\cos(w_0 t + j)$ aplicat la intrarea circuitului liniar descris de functia de transfer H(s) se regaseste la iesire sub forma:

$$y(t)= A\left|H(s)\right|_{s=jw_0}\cos(w_0 t + j + arg\ H(s)_{s=jw_0})$$

adica amplificat cu modulul functiei de transfer la frecventa w_0 si defazat suplimentar cu argumentul functiei de transfer la aceasi frecventa w_0.

Pentru un semnal de intrare format prin sumarea unui numar de semnale sinusoidale:

$$e(t)=A_0+\sum_{k=1}^{N} A_k \cos(w_{0k} t + j_k)$$

iesirea se poate calcula pe baza proprietatii de liniaritate:

$$y(t)= A_0 H(j0)+\sum_{k=1}^{N} A_k \left|H(jw_{0k})\right|\cos(w_{0k} t + j_k + \arg H(jw_{0k}))$$

Obs: Integratorul are in cc amplificare teoretica infinita. Realizat practic un astfel de circuit nu functioneaza. El practic integreaza o componenta continua parazita si se satureaza. Schema a doua este o varianta de realizare practica in care amplificarea la JF a fost limitata. Schema a doua se comporta ca integrator pentru frecvente mult mai mari decit $\alpha 1 = \dfrac{1}{R1\ C}$.

```
> Hs:=-alpha/(s+alpha1);
```

$$Hs := -\frac{\alpha}{s+\alpha 1}$$

```
> Homega:=subs(s=I*omega,Hs);
```

$$Homega := -\frac{\alpha}{I\,\omega+\alpha 1}$$

Atenuarea in cc este:

```
> limit(Homega,omega=0);
```

$$-\frac{\alpha}{\alpha 1}$$

Raspunsul la semnal armonic

In acest caz expresia excitatia $e(t)$ este de forma:

```
> e:=A0*cos(w*t);
```

$$e := A0 \cos(w\,t)$$

Transformata Fourier a excitatie $e(t)$ este:

```
> E:=F[dir](e,t,omega);
```

$$E := A0\,\pi\,\mathrm{Dirac}(\omega + w) + A0\,\pi\,\mathrm{Dirac}(-\omega + w)$$

Transformata Fourier a excitatie $y(t)$ este:

```
> Y:=Homega*E;
```

$$Y := -\frac{\alpha\,(A0\,\pi\,\mathrm{Dirac}(\omega + w) + A0\,\pi\,\mathrm{Dirac}(-\omega + w))}{I\,\omega + \alpha 1}$$

Raspunsul $y(t)$ al circuitului la excitatia $e(t)$ este:

```
> y:=simplify(normal(convert(F[inv](Y,omega,t),trig),expanded));
```

$$y := -\frac{\alpha\,A0\,(w \sin(w\,t) + \alpha 1 \cos(w\,t))}{w^2 + \alpha 1^2}$$

La acelasi rezultat se putea ajunge urmarind pasii de calcul:

$$H(jw) = \frac{a}{jw + a_1}$$

Modulul si faza functiei de transfer:

$$\begin{cases} |H(jw)| = \dfrac{a}{\sqrt{w^2 + a_1^2}} \\ \arg(H(jw)) = p - arctg\dfrac{w}{a_1} \end{cases}$$

Forma generala a raspunsului permanent:

$$y(t) = A_0 |H(jw)| \cos(wt + \arg(H(jw)))$$

Raspunsul permanent este:

$$y(t) = A_0 \frac{a}{\sqrt{w^2 + a_1^2}} \cos\left(wt + p - arctg\frac{w}{a_1} \right)$$

Raspunsul la semnal armonic de frecventa egala cu frecventa polului

```
> e1:=eval(e,w=alpha1);
```

$$e1 := A0 \cos(\alpha 1\,t)$$

```
> y1:=simplify(eval(y,w=alpha1));
```

$$y1 := -\frac{1}{2}\,\frac{\alpha\,A0\,(\sin(\alpha 1\,t) + \cos(\alpha 1\,t))}{\alpha 1}$$

Obs: Semnalul de iesire este defazat fata de intrare cu $-\frac{\pi}{4}$ si atenuat cu $-\frac{\alpha}{\alpha 1 \sqrt{2}}$ (cu $\frac{1}{\sqrt{2}}$ fata de atenuarea din banda).

Raspunsul la semnal armonic de frecventa inalta

```
> e;
```

$$A0 \cos(w\,t)$$

```
> y;
```

$$-\frac{\alpha\, A0\,(w\sin(w\,t)+\alpha 1\cos(w\,t))}{w^2+\alpha 1^2}$$

```
> limit(y,alpha1=0);
```

$$-\frac{\alpha\, A0\sin(w\,t)}{w}$$

Obs: iesirea este integrala intrarii!

Raspuns de regim tranzitoriu

```
> restart:with(inttrans):
> libname:="c:\\maple/SCSlib",libname:
> L:=table([dir=inttrans[laplace],inv=inttrans[invlaplace]]):
>assume(_alpha,positive):assume(_omega,positive):assume(_tau,positive):assume(_T,positive):
```

Consideram circuitul liniar cu AO pentru care am determinat functia de transfer $\mathrm{H}(s)$. Excitatia este $\mathrm{e}(t)$ si raspunsul este tensiunea $\mathrm{y}(t)$. In situatia in care semnalul $\mathrm{e}(t)$ este cauzal, circuitul functioneaza in regim tranzitoriu. In studiul comportarii de regim tranzitoriu se utilizeaza *transformata Laplace*.Vom nota cu $\mathrm{Y}(s)$ transformata Laplace a semnalului $\mathrm{y}(t)$ si cu $\mathrm{E}(s)$ transformata Laplace a semnalului excitatie $\mathrm{e}(t)$. Metodologia de calcul a raspunsului $\mathrm{y}(t)$ la excitatia $\mathrm{e}(t)$ pentru un circuitui este urmatoarea:

- Determinarea lui $\mathrm{E}(s)$ din $\mathrm{e}(t)$, folosind *transformata Laplace directa.*
- Determinarea lui $\mathrm{Y}(s)$, folosind relatia $\mathrm{Y}(s)=\mathrm{H}(s)\,\mathrm{E}(s)$.
- Determinarea lui $\mathrm{y}(t)$ din $\mathrm{Y}(s)$, folosind *transformata Laplace inversa*.

Pentru circuit functia de transfer este:

```
> Hs:=-alpha/(s+alpha1);
```

$$Hs := -\frac{\alpha}{s+\alpha 1}$$

Raspunsul la semnal treapta

In acest caz expresia excitatia $\mathrm{e}(t)$ este de forma:

```
> e:=A0*Heaviside(t);
```

$$e := A0\,\mathrm{Heaviside}(t)$$

- Transformata Laplace a excitatie $\mathrm{e}(t)$ este:

```
> E:=L[dir](e,t,s);
```

$$E := \frac{A0}{s}$$

- Transformata Laplace a excitatie $\mathrm{y}(t)$ este:

```
> Y:=Hs*E;
```

$$Y := -\frac{\alpha\, A0}{(s+\alpha 1)\, s}$$

- Raspunsul $\mathrm{y}(t)$ al circuitului la excitatia $\mathrm{e}(t)$ este:

```
> y:=L[inv](Y,s,t)*Heaviside(t);
```

$$y := -\alpha\, A0 \left(-\frac{e^{(-\alpha 1\, t)}}{\alpha 1} + \frac{1}{\alpha 1} \right) \mathrm{Heaviside}(t)$$

Se reprezita semnalele de intrare si iesire pentru valorile din schema ale rezistentelor si ale condensatorului:

```
> plot(eval([10*e, subs([alpha=1/(R*C), alpha1=1/(R1*C)],y)], [A0=1,C=22*1E-9, R=10^3, R1=10^5]), t=-0.001..0.02,numpoints = 200,thickness=1,axes=box,title="Raspunsul la excitatie data a circuitului cu AO",labels=["t","e(t),y(t)"]);
```



Raspunsul la semnal dreptunghiular

In acest caz expresia excitatia $e(t)$ este de forma:

```
> e:=A0*(Heaviside(t)-Heaviside(t-tau));
```

$$e := A0\,(\mathrm{Heaviside}(t) - \mathrm{Heaviside}(t-\tau))$$

- Transformata Laplace a excitatie $e(t)$ este:

```
> E:=subs(_tau=tau,L[dir](subs(tau=_tau,e),t,s));
```

$$E := A0 \left(\frac{1}{s} - \frac{e^{(-s\,\tau)}}{s} \right)$$

- Transformata Laplace a excitatie $y(t)$ este:

```
> Y:=Hs*E;
```

$$Y := -\frac{\alpha\, A0 \left(\frac{1}{s} - \frac{e^{(-s\,\tau)}}{s} \right)}{s + \alpha 1}$$

- Raspunsul $y(t)$ al circuitului la excitatia $e(t)$ este:

```
> y:=L[inv](Y,s,t)*Heaviside(t);
```

$$y := -\alpha\, A0 \left(\frac{(e^{(\alpha 1\,(-t+\tau))} - 1)\,\mathrm{Heaviside}(t-\tau)}{\alpha 1} - \frac{e^{(-\alpha 1\, t)} - 1}{\alpha 1} \right) \mathrm{Heaviside}(t)$$

Se reprezita semnalele de intrare si iesire pentru valorile din schema ale rezistentelor si ale condensatorului:

```
> plot(eval([10*e, subs([alpha=1/(R*C), alpha1=1/(R1*C)],y)], [A0=1,tau = 0.05, C=22*1E-9, R=10^3, R1=10^5]), t=-0.01..0.1,numpoints = 200,thickness=1,axes=box,title="Raspunsul la excitatie data a circuitului cu AO",labels=["t","e(t),y(t)"]);
```



Raspunsul la succesiune de pulsuri dreptunghiulare, bipolar - simetrice

```
> N:=10:
```

In acest caz expresia excitatia $e(t)$ este de forma:

```
> e:=(-A0/2+A0*sum(Heaviside(t-n*T)-Heaviside(t-tau-n*T),n=0..N-1))*Heaviside(t):
```

- Transformata Laplace a excitatie $e(t)$ este:

```
> E:=subs([_tau=tau,_T=T],L[dir](subs([T=_T,tau=_tau],e),t,s)):
```

- Transformata Laplace a excitatie $y(t)$ este:

```
> Y:=Hs*E:
```

- Raspunsul $y(t)$ al circuitului la excitatia $e(t)$ este:

```
>
y:=subs([_tau=tau,_T=T,_alpha=alpha],L[inv](subs([T=_T,tau=_tau,alpha=_alpha],Y),s,t))*Heaviside(t):
```

Se reprezita semnalele de intrare si iesire pentru valorile din schema ale rezistentelor si ale condensatorului:

```
> plot(eval([10*e, subs([alpha=1/(R*C), alpha1=1/(R1*C)],y)],
[A0=1,T = 0.1, tau = 0.05, C=22*1E-9, R=10^3, R1=10^5]), t=-0.01..0.5,numpoints = 200,thickness=1,axes=box,title="Raspunsul la excitatia data a circuitului cu AO",labels=["t","e(t),y(t)"]);
```



Comportare ca integrator

Se modifica frecventa semnalului dreptunghiular (perioda T) cu pastrarea unui factor de umplere de 50%. Se observa regimul tranzitoriu :

```
> plot(eval([10*e, subs([alpha=1/(R*C), alpha1=1/(R1*C)],y)],
[A0=1,T = 0.002, tau = 0.001, C=22*1E-9, R=10^3, R1=10^5]), t=-0.001..0.02,numpoints =
200,thickness=1,axes=box,title="Raspunsul la excitatia data a
```

```
circuitului cu AO",labels=["t","e(t),y(t)"]);
```



Se vizualizeaza semnalul pe un interval de timp cind regimul tranzitoriu se poate considera stins:

```
> plot(eval([10*e, subs([alpha=1/(R*C), alpha1=1/(R1*C)],y)],
[A0=1,T = 0.002, tau = 0.001, C=22*1E-9, R=10^3, R1=10^5]),
t=0.01..0.02,numpoints =
200,thickness=1,axes=box,title="Raspunsul la excitatia data a
circuitului cu AO",labels=["t","e(t),y(t)"]);
```



Obs : Perioada la care se lucreaza este T<< $\frac{2\pi}{\alpha 1}$.

Analiza PSPICE

Descriere PSpice:

```
*circuit
VIN 1 0 AC 1
VCC+ VCC+ 0 10V
VCC- VCC- 0 -10V
R1 1 21 1K
C1 21 31 22N
XOPAMP1 0 21 VCC+ VCC- 31 uA741
R2 1 22 1K
C2 22 32 22N
Rcom 22 32 100K
XOPAMP2 0 22 VCC+ VCC- 32 uA741
.LIB "opamp.lib"
.AC dec 100 0.1 1000Meg
.PROBE
.END
```

```
*circuit
VIN 1 0 pulse(-1V 1V 0.3mS 1us 2us 0.3ms 0.602ms)
VCC+ VCC+ 0 10V
VCC- VCC- 0 -10V
R1 1 21 1K
C1 21 31 22N
XOPAMP1 0 21 VCC+ VCC- 31 uA741
R2 1 22 1K
C2 22 32 22N
Rcom 22 32 100K
XOPAMP2 0 22 VCC+ VCC- 32 uA741
.LIB "opamp.lib"
.TRAN 5us 150ms 0 10us
.PROBE
.END
```

Diagrama Bode de modul si faza pentru cele doua circuite:



Functia pondere



Comportare de integrator

cu intrare in limitare:



fara intrare in limitare:



stingerea regimului tranzitoriu si amplificare in c.c.